Lesson 10: Polynomials

```
> restart;
```

Factoring polynomials

The **Fundamental Theorem of Algebra** says that a polynomial of degree n in one variable x (with coefficients that are complex numbers) can be written as

$a_n (x - r_1)(x - r_2) \cdots (x - r_n) = a_n \prod_{j=1}^{n} (x - r_j)$ where the complex numbers $r_1, \ldots, r_n$ are the roots of the polynomial. Some roots may be present more than once: these are called **repeated roots**. The number of times a root appears is called its **multiplicity**.

Maple has a **factor** command to factor polynomials, but usually it doesn't do this kind of complete factorization. If the coefficients of the polynomial are rational numbers, it will just find factors with coefficients that are rational numbers. For example:

```
> P:= 4*x^4 -4*x^2 +12*x - 9;
  factor(P);
```

$$P := 4x^4 - 4x^2 + 12x - 9$$

$$(2x^2 + 2x - 3)(2x^2 - 2x + 3) \tag{1.1}$$

To factor completely into linear factors, you can say

```
> factor(P, complex);
```

$$4.\,(x + 1.82287565553230)(x - 0.500000000000000 + 1.11803398874989\,I)(x - 0.500000000000000 - 1.11803398874989\,I)(x - 0.822875655532295) \tag{1.2}$$

Notice that the roots r_j are given as (approximate) floating-point numbers. You could also ask for a factorization over the real numbers:

```
> factor(P, real);
```

$$4.\,(x + 1.82287565553230)(x - 0.822875655532295)(x^2 - 1.\,x + 1.50000000000000) \tag{1.3}$$

This gives you factors where all the coefficients are real: they may be linear factors, or quadratics whose roots are not real.

Multivariate Polynomials

A **multivariate polynomial** is a polynomial with several variables instead of just one. For example:

```
> P := 4*x^4+2*x^3*y-6*x^2*y^2-2*x^2*y-x*y^2+3*y^3+6*x^2+3*x*
  y-9*y^2;
```

$$P := 4x^4 + 2x^3y - 6x^2y^2 - 2x^2y - xy^2 + 3y^3 + 6x^2 + 3xy - 9y^2 \tag{2.1}$$

From one point of view, this can be considered as a polynomial of degree 4 in the variable x whose coefficients are polynomials in y. You could collect the terms of each degree in x as follows:

```
> collect(P,x);
```

$$4x^4 + 2x^3y + (-6y^2 + 6 - 2y)x^2 + (3y - y^2)x - 9y^2 + 3y^3 \tag{2.2}$$

Or you could consider it as a polynomial in y whose coefficients are polynomials in x.

```
> collect(P,y);
```

$$3\,y^3+(-x-6\,x^2-9)\,y^2+(2\,x^3-2\,x^2+3\,x)\,y+4\,x^4+6\,x^2 \quad (2.3)$$

Factoring works (to some extent):

```
> factor(P);
```

$$(3\,y+2\,x)\,(2\,x^2-y+3)\,(-y+x) \quad (2.4)$$

But you can't factor a multivariate polynomial into (multivariate) linear factors. Writing $2\,x^2+3-y=2\left(x+\left(\frac{y-3}{2}\right)^{\left(\frac{1}{2}\right)}\right)\left(x-\left(\frac{y-3}{2}\right)^{\left(\frac{1}{2}\right)}\right)$ doesn't count: these are not polynomials in x and y, because of the non-integer powers.

Here's something Maple can't do: factor multivariate polynomials with floating-point entries. It seems that no good algorithm for this is known.

```
> evalf(P);
```

$$4.\,x^4+2.\,x^3\,y-6.\,x^2\,y^2-2.\,x^2\,y-1.\,x\,y^2+3.\,y^3+6.\,x^2+3.\,x\,y-9.\,y^2 \quad (2.5)$$

```
> factor(%);
```

$$4.000000000\,x^4+2.000000000\,x^3\,y-6.000000000\,x^2\,y^2-2.000000000\,x^2\,y-1.000000000\,x\,y^2+3.000000000\,y^3+6.000000000\,x^2+3.000000000\,x\,y-9.000000000\,y^2 \quad (2.6)$$

```
> convert(%,rational);
```

$$4\,x^4+2\,x^3\,y-6\,x^2\,y^2-2\,x^2\,y-x\,y^2+3\,y^3+6\,x^2+3\,x\,y-9\,y^2 \quad (2.7)$$

```
> factor(%);
```

$$(3\,y+2\,x)\,(2\,x^2-y+3)\,(-y+x) \quad (2.8)$$

Plotting equations in two variables

Here are two polynomials in two variables.

```
> p1 := 2*x^4  - 2*y^3 + y ;
  p2 :=  2 * x^2 * y + 3*y^4 -  2*x ;
```

$$p1 := 2\,x^4-2\,y^3+y$$
$$p2 := 2\,x^2\,y+3\,y^4-2\,x \quad (3.1)$$

We'll want to solve the equations p1 = 0 and p2 = 0. Before doing this, let's plot the curves to see how many intersections they seem to have. Up to now we've plotted expressions such as $f(x)$ in one variable, corresponding to equations $y=f(x)$. It would be hard to do that here: p1 has degree 3 as a polynomial in y and p2 has degree 4, and although those could be solved, the solutions would be extremely complicated and unpleasant. The **plots** package does have a command for plotting curves corresponding to arbitrary equations in two variables: it's called **implicitplot**.

```
> with(plots):
```

```
> implicitplot(p1=0, x = -5 .. 5, y = -5 .. 5);
```



Actually the **=0** is not needed: you could give it an expression instead of an equation, and **implicitplot** would plot the equation (expression) = 0.

```
> implicitplot(p1, x = -5 .. 5, y = -5 .. 5);
```



What **implicitplot** does is basically this. It looks at the difference between the two sides of the equation on a rectangular grid of points in the rectangle where you want the plot: a certain number in the x direction by a certain number in the y direction (these can be set with the **grid** option). Wherever it sees this difference change sign, it knows there is a bit of curve to be drawn, and it draws a straight line segment using a linear interpolation. For example, suppose one cell of the grid consisted of points (0,0), (1,0), (0,1), (1,1), and the difference between the sides of the equation is

```
> f:= (x,y) -> x^2 - y + 1/2;
```

$$f := (x, y) \rightarrow x^2 - y + \frac{1}{2} \quad \textbf{(3.2)}$$

```
> f(0,0), f(1,0), f(0,1), f(1,1);
```

$$\frac{1}{2}, \frac{3}{2}, -\frac{1}{2}, \frac{1}{2} \quad \textbf{(3.3)}$$

The line must go between (0,0) and (0,1) and between (0,1) and (1,1), and in each case linear interpolation says it should go half-way, i.e. a linear function that has the value 1/2 at 0 and -1/2 at 1 would be 0 at 1/2. So it would draw a line segment from [0,1/2] to [1/2, 1].

```
> implicitplot(f(x,y)=0, x=0..1,y=0..1,grid=[2,2], view=[0..1,
  0..1]);
```



One consequence of this strategy is that implicitplot usually doesn't catch a case where the difference of the two sides doesn't change sign, but instead is 0 on some curve and, say, positive on both sides.

```
> implicitplot((y-x^2)^2=0, x=-1..1,y=-1..1);
```



There is one option that may help here: **factor=true**, which tells **implicitplot** to try factoring the input. If it sees that the input is a power f^n, it can work on f instead of f^n.

```
> implicitplot((y-x^2)^2=0,x=-1..1,y=-1..1,factor=true);
```



Implicitplot also has some trouble with curves that intersect themselves, because linear interpolation is usually not very good near the intersection points.

```
> implicitplot((x^2+y^2-1)*((x-1)^2+y^2-1),x=-1..2,y=-1..1);
```



Increasing the grid settings may help somewhat, but it makes the command slower.
By the way, the option **scaling = constrained** makes a plot have the same scale in both axes, so circles look like circles and not ellipses.

```
> implicitplot((x^2+y^2-1)*((x-1)^2+y^2-1),x=-1..2,y=-1..1,
  grid=[50,50],scaling=constrained);
```



There are also two options: **crossingrefine** and **gridrefine**, which you can set to positive integer values. These cause **implicitplot** to look more closely at what happens near the points it plots, giving you better-looking curves.`

```
> implicitplot((x^2+y^2-1)*((x-1)^2+y^2-1),x=-1..2,y=-1..1,
  gridrefine=3, scaling=constrained);
```



You can also give implicitplot a list of equations rather than just one, and use different colours for each. So here are our two equations.

```
> implicitplot([p1 = 0, p2 = 0], x = -5 .. 5, y = -5 .. 5,
  colour = [red,blue], grid=[50,50],gridrefine=3);
```



It looks very much like there are two intersections, one at the origin and another near (0.28, -0.68), although you might want to take a closer look near (0.38, 0.72).

Solving systems

As we've seen, we can ask Maple to solve this system of equations for the two variables *x* and *y*. We could try either **fsolve** or **solve**. On a system of equations, **fsolve** will only return one solution (it's just for a single polynomial that it would return all the real solutions).

```
> fsolve({p1=0,p2=0},{x,y});
```

$$\{x = 0.2968882966,\ y = -0.6992057186\} \quad \textbf{(4.1)}$$

If we want another solution, we can give **fsolve** intervals for *x* and *y*, making sure not to include the solution it already found.

```
> fsolve({p1=0,p2=0},{x=-1..1,y=-0.5 .. 1});
```

$$\{x = 0.,\ y = 0.\} \quad \textbf{(4.2)}$$

```
> fsolve({p1=0,p2=0},{x=0 .. 0.5, y = 0.5 .. 1});
```

$$\mathit{fsolve}\left(\{2\,x^4 - 2\,y^3 + y = 0,\ 2\,x^2\,y + 3\,y^4 - 2\,x = 0\},\ \{x, y\},\ \{x = 0\,..0.5,\ y = 0.5\,..1\}\right) \quad \textbf{(4.3)}$$

There is another command, **Isolate** in the **RootFinding** package, that will give you all the real solutions of a system of polynomials.

```
> RootFinding[Isolate]([p1,p2],[x,y]);
```

$$[[x=0.2968882966, y=-0.6992057186], [x=0., y=0.]] \quad (4.4)$$

It wouldn't work for non-polynomials, though.

```
> RootFinding[Isolate]([x^2 - y*cos(x), x-y],[x,y]);
Error, (in RootFinding:-Isolate) symbolic coefficients are not supported, {cos(x)}
```

We could also try **solve**:

```
> solve({p1 = 0, p2 = 0}, {x,y});
```

$$\{x=0, y=0\}, \{x=(\mathit{RootOf}(81\_Z^{15}-72\_Z^{12}+36\_Z^{10}+16\_Z^{9}+80\_Z^{7}-44\_Z^{5}-16\_Z^{2}+8)^{4}(-4\,\mathit{RootOf}(81\_Z^{15}-72\_Z^{12}+36\_Z^{10}+16\_Z^{9}+80\_Z^{7}-44\_Z^{5}-16\_Z^{2}+8)^{2}-4+9\,\mathit{RootOf}(81\_Z^{15}-72\_Z^{12}+36\_Z^{10}+16\_Z^{9}+80\_Z^{7}-44\_Z^{5}-16\_Z^{2}+8)^{5}))/(-4+12\,\mathit{RootOf}(81\_Z^{15}-72\_Z^{12}+36\_Z^{10}+16\_Z^{9}+80\_Z^{7}-44\_Z^{5}-16\_Z^{2}+8)^{5}), y=\mathit{RootOf}(81\_Z^{15}-72\_Z^{12}+36\_Z^{10}+16\_Z^{9}+80\_Z^{7}-44\_Z^{5}-16\_Z^{2}+8)\} \quad (4.5)$$

You can also use lists instead of sets. The main advantage of using lists instead of sets here (I think) is that, since lists have a well-defined order, you should get a more predictable result (the same variable should always be eliminated first).

```
> S := solve([p1 = 0, p2 = 0], [x,y]);
```

$$S := [[x=0, y=0], [x=(\mathit{RootOf}(81\_Z^{15}-72\_Z^{12}+36\_Z^{10}+16\_Z^{9}+80\_Z^{7}-44\_Z^{5}-16\_Z^{2}+8)^{4}(-4\,\mathit{RootOf}(81\_Z^{15}-72\_Z^{12}+36\_Z^{10}+16\_Z^{9}+80\_Z^{7}-44\_Z^{5}-16\_Z^{2}+8)^{2}-4+9\,\mathit{RootOf}(81\_Z^{15}-72\_Z^{12}+36\_Z^{10}+16\_Z^{9}+80\_Z^{7}-44\_Z^{5}-16\_Z^{2}+8)^{5}))/(-4+12\,\mathit{RootOf}(81\_Z^{15}-72\_Z^{12}+36\_Z^{10}+16\_Z^{9}+80\_Z^{7}-44\_Z^{5}-16\_Z^{2}+8)^{5}), y=\mathit{RootOf}(81\_Z^{15}-72\_Z^{12}+36\_Z^{10}+16\_Z^{9}+80\_Z^{7}-44\_Z^{5}-16\_Z^{2}+8)]] \quad (4.6)$$

It did find the solutions after a fashion: y is "RootOf" a complicated polynomial, and x is a very complicated function of that. It's our bad luck that the first complicated polynomial can't be solved in "closed form".

To get all the values of a RootOf (or an expression containing a RootOf), you can use **allvalues**. In this case we want to apply that to the second solution returned by **solve**.

```
> allvalues(S[2]);
```

$$[x=(\mathit{RootOf}(81\_Z^{15}-72\_Z^{12}+36\_Z^{10}+16\_Z^{9}+80\_Z^{7}-44\_Z^{5}-16\_Z^{2}+8, \mathit{index}=1)^{4}(-4\,\mathit{RootOf}(81\_Z^{15}-72\_Z^{12}+36\_Z^{10}+16\_Z^{9}+80\_Z^{7}-44\_Z^{5}-16\_Z^{2}+8, \mathit{index}=1)^{2}-4+9\,\mathit{RootOf}(81\_Z^{15}-72\_Z^{12}+36\_Z^{10}+16\_Z^{9}+80\_Z^{7}-44\_Z^{5}-16\_Z^{2}+8, \mathit{index}=1)^{5}))/(-4+12\,\mathit{RootOf}(81\_Z^{15}-72\_Z^{12}+36\_Z^{10}+16\_Z^{9}+80\_Z^{7}-44\_Z^{5}-16\_Z^{2}+8, \mathit{index}=1)^{5}), y=\mathit{RootOf}(81\_Z^{15}-72\_Z^{12}+36\_Z^{10}+16\_Z^{9}+80\_Z^{7}-44\_Z^{5}-16\_Z^{2}+8, \mathit{index} \quad (4.7)$$

$= 1)], [x = (RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 2)^4 (-4\, RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 2)^2 - 4 + 9\, RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 2)^5)) / (-4 + 12\, RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 2)^5), y = RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 2)], [x = (RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 3)^4 (-4\, RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 3)^2 - 4 + 9\, RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 3)^5)) / (-4 + 12\, RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 3)^5), y = RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 3)], [x = (RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 4)^4 (-4\, RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 4)^2 - 4 + 9\, RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 4)^5)) / (-4 + 12\, RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 4)^5), y = RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 4)], [x = (RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 5)^4 (-4\, RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 5)^2 - 4 + 9\, RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 5)^5)) / (-4 + 12\, RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 5)^5), y = RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 5)], [x = (RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 6)^4 (-4\, RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 6)^2 - 4 + 9\, RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2 + 8, index = 6)^5)) / (-4 + 12\, RootOf(81\_Z^{15} - 72\_Z^{12} + 36\_Z^{10} + 16\_Z^9 + 80\_Z^7 - 44\_Z^5 - 16\_Z^2$

$+ 8, \mathit{index} = 6)^{5}), y = \mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12} + 36 \_Z^{10} + 16 \_Z^{9} + 80 \_Z^{7} - 44 \_Z^{5}$

$- 16 \_Z^{2} + 8, \mathit{index} = 6)], [x = (\mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12} + 36 \_Z^{10} + 16 \_Z^{9}$

$+ 80 \_Z^{7} - 44 \_Z^{5} - 16 \_Z^{2} + 8, \mathit{index} = 7)^{4} (-4 \mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12}$

$+ 36 \_Z^{10} + 16 \_Z^{9} + 80 \_Z^{7} - 44 \_Z^{5} - 16 \_Z^{2} + 8, \mathit{index} = 7)^{2} - 4$

$+ 9 \mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12} + 36 \_Z^{10} + 16 \_Z^{9} + 80 \_Z^{7} - 44 \_Z^{5} - 16 \_Z^{2}$

$+ 8, \mathit{index} = 7)^{5})) / (-4 + 12 \mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12} + 36 \_Z^{10} + 16 \_Z^{9}$

$+ 80 \_Z^{7} - 44 \_Z^{5} - 16 \_Z^{2} + 8, \mathit{index} = 7)^{5}), y = \mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12} + 36 \_Z^{10}$

$+ 16 \_Z^{9} + 80 \_Z^{7} - 44 \_Z^{5} - 16 \_Z^{2} + 8, \mathit{index} = 7)], [x = (\mathit{RootOf}(81 \_Z^{15}$

$- 72 \_Z^{12} + 36 \_Z^{10} + 16 \_Z^{9} + 80 \_Z^{7} - 44 \_Z^{5} - 16 \_Z^{2} + 8, \mathit{index} = 8)^{4} ($

$-4 \mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12} + 36 \_Z^{10} + 16 \_Z^{9} + 80 \_Z^{7} - 44 \_Z^{5} - 16 \_Z^{2} + 8,$

$\mathit{index} = 8)^{2} - 4 + 9 \mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12} + 36 \_Z^{10} + 16 \_Z^{9} + 80 \_Z^{7}$

$- 44 \_Z^{5} - 16 \_Z^{2} + 8, \mathit{index} = 8)^{5})) / (-4 + 12 \mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12}$

$+ 36 \_Z^{10} + 16 \_Z^{9} + 80 \_Z^{7} - 44 \_Z^{5} - 16 \_Z^{2} + 8, \mathit{index} = 8)^{5}), y = \mathit{RootOf}(81 \_Z^{15}$

$- 72 \_Z^{12} + 36 \_Z^{10} + 16 \_Z^{9} + 80 \_Z^{7} - 44 \_Z^{5} - 16 \_Z^{2} + 8, \mathit{index} = 8)], [x$

$= (\mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12} + 36 \_Z^{10} + 16 \_Z^{9} + 80 \_Z^{7} - 44 \_Z^{5} - 16 \_Z^{2} + 8,$

$\mathit{index} = 9)^{4} (-4 \mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12} + 36 \_Z^{10} + 16 \_Z^{9} + 80 \_Z^{7} - 44 \_Z^{5}$

$- 16 \_Z^{2} + 8, \mathit{index} = 9)^{2} - 4 + 9 \mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12} + 36 \_Z^{10} + 16 \_Z^{9}$

$+ 80 \_Z^{7} - 44 \_Z^{5} - 16 \_Z^{2} + 8, \mathit{index} = 9)^{5})) / (-4 + 12 \mathit{RootOf}(81 \_Z^{15}$

$- 72 \_Z^{12} + 36 \_Z^{10} + 16 \_Z^{9} + 80 \_Z^{7} - 44 \_Z^{5} - 16 \_Z^{2} + 8, \mathit{index} = 9)^{5}), y$

$= \mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12} + 36 \_Z^{10} + 16 \_Z^{9} + 80 \_Z^{7} - 44 \_Z^{5} - 16 \_Z^{2} + 8, \mathit{index}$

$= 9)], [x = (\mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12} + 36 \_Z^{10} + 16 \_Z^{9} + 80 \_Z^{7} - 44 \_Z^{5}$

$- 16 \_Z^{2} + 8, \mathit{index} = 10)^{4} (-4 \mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12} + 36 \_Z^{10} + 16 \_Z^{9}$

$+ 80 \_Z^{7} - 44 \_Z^{5} - 16 \_Z^{2} + 8, \mathit{index} = 10)^{2} - 4 + 9 \mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12}$

$+ 36 \_Z^{10} + 16 \_Z^{9} + 80 \_Z^{7} - 44 \_Z^{5} - 16 \_Z^{2} + 8, \mathit{index} = 10)^{5})) / (-4$

$+ 12 \mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12} + 36 \_Z^{10} + 16 \_Z^{9} + 80 \_Z^{7} - 44 \_Z^{5} - 16 \_Z^{2}$

$+ 8, \mathit{index} = 10)^{5}), y = \mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12} + 36 \_Z^{10} + 16 \_Z^{9} + 80 \_Z^{7}$

$- 44 \_Z^{5} - 16 \_Z^{2} + 8, \mathit{index} = 10)], [x = (\mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12} + 36 \_Z^{10}$

$+ 16 \_Z^{9} + 80 \_Z^{7} - 44 \_Z^{5} - 16 \_Z^{2} + 8, \mathit{index} = 11)^{4} (-4 \mathit{RootOf}(81 \_Z^{15}$

$- 72 \_Z^{12} + 36 \_Z^{10} + 16 \_Z^{9} + 80 \_Z^{7} - 44 \_Z^{5} - 16 \_Z^{2} + 8, \mathit{index} = 11)^{2} - 4$

$+ 9 \mathit{RootOf}(81 \_Z^{15} - 72 \_Z^{12} + 36 \_Z^{10} + 16 \_Z^{9} + 80 \_Z^{7} - 44 \_Z^{5} - 16 \_Z^{2}$

$$+ 8, \mathit{index} = 11)^{5}))/(-4 + 12\, \mathit{RootOf}(81\, \_Z^{15} - 72\, \_Z^{12} + 36\, \_Z^{10} + 16\, \_Z^{9}$$
$$+ 80\, \_Z^{7} - 44\, \_Z^{5} - 16\, \_Z^{2} + 8, \mathit{index} = 11)^{5}), y = \mathit{RootOf}(81\, \_Z^{15} - 72\, \_Z^{12}$$
$$+ 36\, \_Z^{10} + 16\, \_Z^{9} + 80\, \_Z^{7} - 44\, \_Z^{5} - 16\, \_Z^{2} + 8, \mathit{index} = 11)], [x$$
$$= (\mathit{RootOf}(81\, \_Z^{15} - 72\, \_Z^{12} + 36\, \_Z^{10} + 16\, \_Z^{9} + 80\, \_Z^{7} - 44\, \_Z^{5} - 16\, \_Z^{2} + 8,$$
$$\mathit{index} = 12)^{4} (-4\, \mathit{RootOf}(81\, \_Z^{15} - 72\, \_Z^{12} + 36\, \_Z^{10} + 16\, \_Z^{9} + 80\, \_Z^{7} - 44\, \_Z^{5}$$
$$- 16\, \_Z^{2} + 8, \mathit{index} = 12)^{2} - 4 + 9\, \mathit{RootOf}(81\, \_Z^{15} - 72\, \_Z^{12} + 36\, \_Z^{10} + 16\, \_Z^{9}$$
$$+ 80\, \_Z^{7} - 44\, \_Z^{5} - 16\, \_Z^{2} + 8, \mathit{index} = 12)^{5}))/(-4 + 12\, \mathit{RootOf}(81\, \_Z^{15}$$
$$- 72\, \_Z^{12} + 36\, \_Z^{10} + 16\, \_Z^{9} + 80\, \_Z^{7} - 44\, \_Z^{5} - 16\, \_Z^{2} + 8, \mathit{index} = 12)^{5}), y$$
$$= \mathit{RootOf}(81\, \_Z^{15} - 72\, \_Z^{12} + 36\, \_Z^{10} + 16\, \_Z^{9} + 80\, \_Z^{7} - 44\, \_Z^{5} - 16\, \_Z^{2} + 8, \mathit{index}$$
$$= 12)], [x = (\mathit{RootOf}(81\, \_Z^{15} - 72\, \_Z^{12} + 36\, \_Z^{10} + 16\, \_Z^{9} + 80\, \_Z^{7} - 44\, \_Z^{5}$$
$$- 16\, \_Z^{2} + 8, \mathit{index} = 13)^{4} (-4\, \mathit{RootOf}(81\, \_Z^{15} - 72\, \_Z^{12} + 36\, \_Z^{10} + 16\, \_Z^{9}$$
$$+ 80\, \_Z^{7} - 44\, \_Z^{5} - 16\, \_Z^{2} + 8, \mathit{index} = 13)^{2} - 4 + 9\, \mathit{RootOf}(81\, \_Z^{15} - 72\, \_Z^{12}$$
$$+ 36\, \_Z^{10} + 16\, \_Z^{9} + 80\, \_Z^{7} - 44\, \_Z^{5} - 16\, \_Z^{2} + 8, \mathit{index} = 13)^{5}))/(-4$$
$$+ 12\, \mathit{RootOf}(81\, \_Z^{15} - 72\, \_Z^{12} + 36\, \_Z^{10} + 16\, \_Z^{9} + 80\, \_Z^{7} - 44\, \_Z^{5} - 16\, \_Z^{2}$$
$$+ 8, \mathit{index} = 13)^{5}), y = \mathit{RootOf}(81\, \_Z^{15} - 72\, \_Z^{12} + 36\, \_Z^{10} + 16\, \_Z^{9} + 80\, \_Z^{7}$$
$$- 44\, \_Z^{5} - 16\, \_Z^{2} + 8, \mathit{index} = 13)], [x = (\mathit{RootOf}(81\, \_Z^{15} - 72\, \_Z^{12} + 36\, \_Z^{10}$$
$$+ 16\, \_Z^{9} + 80\, \_Z^{7} - 44\, \_Z^{5} - 16\, \_Z^{2} + 8, \mathit{index} = 14)^{4} (-4\, \mathit{RootOf}(81\, \_Z^{15}$$
$$- 72\, \_Z^{12} + 36\, \_Z^{10} + 16\, \_Z^{9} + 80\, \_Z^{7} - 44\, \_Z^{5} - 16\, \_Z^{2} + 8, \mathit{index} = 14)^{2} - 4$$
$$+ 9\, \mathit{RootOf}(81\, \_Z^{15} - 72\, \_Z^{12} + 36\, \_Z^{10} + 16\, \_Z^{9} + 80\, \_Z^{7} - 44\, \_Z^{5} - 16\, \_Z^{2}$$
$$+ 8, \mathit{index} = 14)^{5}))/(-4 + 12\, \mathit{RootOf}(81\, \_Z^{15} - 72\, \_Z^{12} + 36\, \_Z^{10} + 16\, \_Z^{9}$$
$$+ 80\, \_Z^{7} - 44\, \_Z^{5} - 16\, \_Z^{2} + 8, \mathit{index} = 14)^{5}), y = \mathit{RootOf}(81\, \_Z^{15} - 72\, \_Z^{12}$$
$$+ 36\, \_Z^{10} + 16\, \_Z^{9} + 80\, \_Z^{7} - 44\, \_Z^{5} - 16\, \_Z^{2} + 8, \mathit{index} = 14)], [x$$
$$= (\mathit{RootOf}(81\, \_Z^{15} - 72\, \_Z^{12} + 36\, \_Z^{10} + 16\, \_Z^{9} + 80\, \_Z^{7} - 44\, \_Z^{5} - 16\, \_Z^{2} + 8,$$
$$\mathit{index} = 15)^{4} (-4\, \mathit{RootOf}(81\, \_Z^{15} - 72\, \_Z^{12} + 36\, \_Z^{10} + 16\, \_Z^{9} + 80\, \_Z^{7} - 44\, \_Z^{5}$$
$$- 16\, \_Z^{2} + 8, \mathit{index} = 15)^{2} - 4 + 9\, \mathit{RootOf}(81\, \_Z^{15} - 72\, \_Z^{12} + 36\, \_Z^{10} + 16\, \_Z^{9}$$
$$+ 80\, \_Z^{7} - 44\, \_Z^{5} - 16\, \_Z^{2} + 8, \mathit{index} = 15)^{5}))/(-4 + 12\, \mathit{RootOf}(81\, \_Z^{15}$$
$$- 72\, \_Z^{12} + 36\, \_Z^{10} + 16\, \_Z^{9} + 80\, \_Z^{7} - 44\, \_Z^{5} - 16\, \_Z^{2} + 8, \mathit{index} = 15)^{5}), y$$
$$= \mathit{RootOf}(81\, \_Z^{15} - 72\, \_Z^{12} + 36\, \_Z^{10} + 16\, \_Z^{9} + 80\, \_Z^{7} - 44\, \_Z^{5} - 16\, \_Z^{2} + 8, \mathit{index}$$
$$= 15)]$$

All that really happened here was that the solution using RootOf was expanded into 15 such solutions, each one where the RootOf was given an index 1 to 15. Note that the result of **allvalues** is an expression sequence. In order to get something easier to work with, we'll make it into a list by enclosing it in square brackets. Next we use **evalf** to get numerical values.

```
> S2:= evalf([%]);
```

$S2 := [[x = 0.4088835987 + 0.2213222308\,I, y = 0.6921985873 + 0.04574550742\,I], [x = -0.3867033783 + 0.8848037969\,I, y = 0.9644222739 + 0.4081484176\,I], [x = 0.7966066723 - 0.5214024214\,I, y = 0.2723934785 + 0.7358796527\,I], [x = -0.5214433312 - 0.8482128195\,I, y = 0.2420756341 + 0.8090287926\,I], [x = 0.1792182542 + 1.189433167\,I, y = -0.4485184522 + 1.089052952\,I], [x = -0.8166124849 - 0.08313922143\,I, y = -0.6059866650 + 0.4303097517\,I], [x = 0.1916065196 - 0.8707008718\,I, y = -0.7669819973 + 0.4552090965\,I], [x = 0.2968882966, y = -0.6992057186], [x = 0.1916065196 + 0.8707008718\,I, y = -0.7669819973 - 0.4552090965\,I], [x = -0.8166124849 + 0.08313922143\,I, y = -0.6059866650 - 0.4303097517\,I], [x = 0.1792182542 - 1.189433167\,I, y = -0.4485184522 - 1.089052952\,I], [x = -0.5214433312 + 0.8482128195\,I, y = 0.2420756341 - 0.8090287926\,I], [x = 0.7966066723 + 0.5214024214\,I, y = 0.2723934785 - 0.7358796527\,I], [x = -0.3867033783 - 0.8848037969\,I, y = 0.9644222739 - 0.4081484176\,I], [x = 0.4088835987 - 0.2213222308\,I, y = 0.6921985873 - 0.04574550742\,I]]$ **(4.8)**

Most of the solutions are complex. We just want the real ones (those with no I).

```
> remove(has, S2, I);
```

$$[[x = 0.2968882966, y = -0.6992057186]]$$ **(4.9)**

Here's what's going on in this bit of Maple magic.
has(A, B) checks whether a Maple expression **A** contains a certain subexpression **B**, returning either *true* or *false*. For example:

```
> has(x^2 + y, x);
```

true **(4.10)**

```
> has(x^2 + y^3, y^2);
```

false **(4.11)**

remove(f, A) takes a function **f** (which should return either *true* or *false*), applies it to each of the operands of **A** (e.g. the members of a list or set) and removes those where the functions return *true,* keeping those where it returns *false.*
select(f, A) does the opposite, keeping those where the function returns *true.*
Additional inputs to **f** can be included in the call to **remove** or **select** after the **A**.
Thus in our case, **remove(has, S2, I)** will evaluate **has(t, I)** for each of the members of the list S2, and returns the list consisting of those members **t** for which the result was *false,* i.e. those that contain only real numbers.

Resultants

What is that polynomial that *y* is supposed to be a root of?

```
> res := resultant(p1,p2,x); factor(res);
```

$$res := 324\,y^{16} - 288\,y^{13} + 144\,y^{11} + 64\,y^{10} + 320\,y^{8} - 176\,y^{6} - 64\,y^{3} + 32\,y$$

$$4\,y\,(81\,y^{15} - 72\,y^{12} + 36\,y^{10} + 16\,y^{9} + 80\,y^{7} - 44\,y^{5} - 16\,y^{2} + 8) \quad \textbf{(5.1)}$$

This is the **resultant** of *p1* and *p2*, considered as polynomials in x (with coefficients that are polynomials in y). The most important property is that if two polynomials in y have a common root, their resultant is 0.

To understand resultants, it's simplest to eliminate the distraction of the second variable, and start with polynomials with constant coefficients. I'll take these same polynomials, but fix $y = 3$.

```
> f1 := eval(p1, y=3);
  f2 := eval(p2, y=3);
```

$$f1 := 2\,x^4 - 51$$

$$f2 := 6\,x^2 + 243 - 2\,x \quad \textbf{(5.2)}$$

```
> resultant(f1,f2,x);
```

$$13519230468 \quad \textbf{(5.3)}$$

Suppose $f_1(x) = a\,x^n + \ldots$ and $f_2(x) = b\,x^m + \ldots$ are polynomials of degrees n and m respectively. In our example, $a = 2$, $n = 4$, $b = 6$, $m = 2$.

Let the n roots of $f_1(x)$ be $r_1, \ldots, r_n$ and the m roots of $f_2(x)$ be $s_1, \ldots, s_m$ (including repeated roots, if any). Then the **resultant** of f_1 and f_2 is $a^m\,b^n \left(\prod_{j=1}^{m}\left(\prod_{i=1}^{n}(r_i - s_j)\right)\right)$

The symbol $\prod$ stands for product. We need to multiply all the differences $r_i - s_j$ where r_i is a root of $f_1(x)$ and s_j is a root of $f_2(x)$, and finally multiply the result by $a^m b^n$.

Note that this will be 0 if f_1 and f_2 have a root in common.

Let's calculate this for our example, using this definition. First we need the roots.

```
> Digits:= 16:
  r := fsolve(f1, x, complex);
  s := fsolve(f2, x, complex);
```

$r :=$ -2.247165429865153, -2.247165429865153 I, 2.247165429865153 I, 2.247165429865153

$s :=$ 0.1666666666666667 − 6.361778227997438 I, 0.1666666666666667 + 6.361778227997438 I **(5.4)**

OK, four roots for f1 and two for f2. Now to plug in to the formula for the resultant.

The **mul** command will make a sequence of expressions, depending on an index variable, and multiply them together. For example, for $\prod_{i=1}^{3} c_i$:

```
> mul(c[i],i=1..3);
```

$$c_1\,c_2\,c_3 \quad \textbf{(5.5)}$$

By the way, for addition we would use **add**.

```
> add(c[i],i=1..3);
```

$$c_1 + c_2 + c_3 \quad \textbf{(5.6)}$$

In this case I need one **mul** inside another.

```
> 2^2 * 6^4 * mul(mul(r[i]-s[j], j=1..2), i=1..4);
```

$$1.351923046800000\ 10^{10} + 3.438906887494981\ 10^{-8}\, I \qquad (5.7)$$

```
> round(%);
```

$$13519230468 \qquad (5.8)$$

Up to roundoff error, this is the same as what Maple got for the resultant.

Resultant using division

If that was all there was to computing the resultant, it would be rather useless. Fortunately, there are algebraic ways to calculate it, which don't depend on actually getting the roots, just using the coefficients. This is why it is useful, especially for polynomials in several variables.

Note that $f_2(x) = b\left(\prod_{j=1}^{m}(x - s_j)\right)$

so the resultant of f_1 and f_2 must be $a^m\left(\prod_{i=1}^{n} f_2(r_i)\right)$

Similarly, the resultant is also $(-1)^{m\,n}\, b^n\left(\prod_{j=1}^{m} f_1(s_j)\right)$

where the $(-1)^{m\,n}$ comes from the fact that each $s_j - r_i$ must be changed to $r_i - s_j$. Let's try these in our example.

```
> 2^2 * mul(eval(f2,x=r[i]), i=1..4);
```

$$1.351923046800000\ 10^{10} - 7.155308202911300\ 10^{-8}\, I \qquad (6.1)$$

```
> (-1)^8 * 6^4 * mul(eval(f1,x=s[j]), j=1..2);
```

$$1.351923046800000\ 10^{10} + 0.\, I \qquad (6.2)$$

This would still not be useful if we needed to calculate the roots of one of our polynomials. But now we use division with remainder:

Given any two polynomials $f_1(x)$ and $f_2(x)$, $f_2(x) \neq 0$, we can write $\boldsymbol{f_1(x) = Q(x)\, f_2(x) + f_3(x)}$ **where $\boldsymbol{Q(x)}$ and $\boldsymbol{f_3(x)}$ are polynomials, and $\boldsymbol{f_3(x)}$ has lower degree than $\boldsymbol{f_2(x)}$.** We can get $Q(x)$ and $f_3(x)$ using **quo** and **rem**.

```
> Q:= quo(f1,f2,x);
  f3 := rem(f1,f2,x);
```

$$Q := \frac{1}{3}x^2 + \frac{1}{9}x - \frac{727}{54}$$

$$f3 := \frac{6441}{2} - \frac{1456}{27}x \qquad (6.3)$$

Now the key observation is that $f_1(s_i) = f_3(s_i)$, and so (except for the factor $(-1)^{m\,n}\, b^m$, which will have to be adjusted) the resultant of f_1 and f_2 is the same as the resultant of f_3 and f_2.

```
> (-1)^8 * 6^4 * mul(eval(f3,x=s[j]), j=1..2);
```

(6.4)

$$1.351923046800000\ 10^{10} + 0.\ I \quad \textbf{(6.4)}$$

```
> (-1)^2 * 6^1 * mul(eval(f1,x=s[j]), j=1..2) = evalf(resultant
  (f3,f2,x));
```

$$6.258902994444444\ 10^{7} + 0.\ I = 6.258902994444444\ 10^{7} \quad \textbf{(6.5)}$$

```
> resultant(f1,f2,x) = (-1)^(8-2)*6^(4-1)*resultant(f3,f2,x);
```

$$13519230468 = 13519230468 \quad \textbf{(6.6)}$$

So instead of calculating the resultant of f_1 (of degree 4) and f_2 (degree 2), we only have f_3 (of degree 1) and f_2 to deal with. Next, take the remainder in dividing f_2 by f_3.

```
> f4 := rem(f2, f3, x);
```

$$f4 := \frac{91254805659}{4239872} \quad \textbf{(6.7)}$$

```
> resultant(f3,f2,x)=(1456/27)^2*resultant(f3,f4,x);
```

$$\frac{1126602539}{18} = \frac{1126602539}{18} \quad \textbf{(6.8)}$$

Now f_4 is just a number, so it has no roots at all. The products are just 1, so the resultant of f_3 and f_4 is $f_4^{degree(f_3)}$.

```
> resultant(f3,f4,x) = (-1456/27)^0*f4^1;
```

$$\frac{91254805659}{4239872} = \frac{91254805659}{4239872} \quad \textbf{(6.9)}$$

Maple objects introduced in this lesson

collect
implicitplot
grid
scaling=constrained
crossingrefine
gridrefine
allvalues
has
remove
select
resultant
mul
add
quo
rem